
Formális nyelvek a gyakorlatban
Formális nyelvek, 1. gyakorlat

Célja: A programozási nyelvek szintaxisának leírására használatos eszközök,
módszerek bemutatása

Fogalmak: BNF, szabály, levezetés, leírt nyelv, EBNF, szintaxis-diagram,
EP programozási nyelv, Pascal programozási nyelv

Feladatok jellege: Egyszerűbb programnyelvi konstrukciók BNF leírása, majd átírása
EBNF-re, az EP megadása EBNF-el (előre kiadott lapon), program készítés EP-ben,
a PASCAL szintaxis-gráfokon alapuló leírásának áttekintése (előre kiadott lapokon),
bizonyos konstrukciók átírása EBNF-re.

BNF
BNF definíciója

BNF: Backus-Naur Form
Metanyelv, melynek segítségével szabályok alkothatók meg.

Építőkövei:

• 〈név〉 (fogalmak)

• ::= a szabály bal- és jobboldalának elválasztászára szolgál.

• a definiálandó nyelv karakterkészlete (terminálisok)

Szabályok

Egy szabály bal- és jobboldalból áll, köztük ::=, baloldalon pontosan 1
fogalom, jobboldalon terminálisok és fogalmak sorozata lehet.

Példa BNF-re:〈azonosító〉
Szabályok

〈azonosító〉 ::= 〈betű〉
〈azonosító〉 ::= 〈betű〉〈azonosítóvég〉
〈azonosítóvég〉 ::= 〈betű〉
〈azonosítóvég〉 ::= 〈számjegy〉
〈azonosítóvég〉 ::= 〈betű〉〈azonosítóvég〉
〈azonosítóvég〉 ::= 〈számjegy〉〈azonosítóvég〉
〈betű〉 ::=a
...
〈betű〉 ::=z
〈számjegy〉 ::=0
...
〈számjegy〉 ::=9

Példa BNF-re:〈azonosító〉
Levezetés

Milyen értelemben ír le azonosítót?

Pl. a12 azonosító-e?

〈azonosító〉→〈betű〉〈azonosítóvég〉→
a〈azonosítóvég〉→a〈számjegy〉〈azonosítóvég〉→
a1〈azonosítóvég〉→a1〈számjegy〉→a12

Ekkor azt mondjuk, hogy az 〈azonosító〉 fogalomból levezethető a12.

u helyes azonosító ⇔ u előállítható az 〈azonosító〉 fogalomból a fenti
helyettesítési mechanizmussal.

Példa BNF-re:〈azonosító〉
Csoportosítás

Egyszerűbb írasmód: alternatívák, kiemelések (csoport)

〈azonosító〉 ::= 〈betű〉|〈betű〉〈azonosítóvég〉
〈azonosító〉 ::= 〈betű〉{ |〈azonosítóvég〉}
〈azonosítóvég〉 ::=〈betű〉|〈számjegy〉|〈betű〉〈azonosítóvég〉|

〈számjegy〉〈azonosítóvég〉
〈azonosítóvég〉 ::={〈betű〉|〈számjegy〉}{ |〈azonosítóvég〉}
〈betű〉 ::=a|b| . . . z
〈számjegy〉 ::=0|1| . . . 9

Az ismétlődéseket a BNF rekurzióval kezeli!
Pl. 〈azonosítóvég〉

EBNF
BNF kiterjesztése

EBNF: Extended Backus-Naur Form
iteratív változat

〈azonosító〉 ::= 〈betű〉@{〈betű〉|〈számjegy〉}
akárhány lépésű iteráció.

〈azonosító〉 ::= 〈betű〉{〈betű〉|〈számjegy〉}5
0

legfeljebb 5, legalább 0 ismétlődés.

EBNF előnyei
Példa: Valós számok BNF-fel

〈törtszám〉 ::= 〈egész〉.|〈egész〉.〈egész〉|.〈egész〉
〈egész〉 ::= 〈számjegy〉|〈számjegy〉〈egész〉
〈számjegy〉 ::=0|1| · · · |9
〈előjel〉 ::= |+ |−
〈valósszám〉 ::=〈egész〉E〈számjegy〉|

〈egész〉E〈előjel〉〈számjegy〉〈számjegy〉|
〈törtszám〉|
〈törtszám〉E〈előjel〉〈számjegy〉|
〈törtszám〉E〈előjel〉〈számjegy〉〈számjegy〉

EBNF előnyei
Példa: Valós számokEBNF-fel

〈törtszám〉 ::= 〈egész〉.{|〈egész〉}|.〈egész〉
〈egész〉 ::= 〈számjegy〉@〈számjegy〉
〈számjegy〉 ::=0|1| · · · |9
〈valósszám〉 ::= 〈törtszám〉|
{〈törtszám〉|〈egész〉}E{ |+ |−}〈számjegy〉{ |〈számjegy〉}

Az EP nyelv leírása EBNF-fel/1

〈program〉 ::= 〈utasítás〉;@{〈utasítás〉;}
〈utasítás〉 ::={〈címke〉:| }〈törzs〉
〈törzs〉 ::={READ|WRITE}〈azonosító〉|

LET 〈azonosító〉=〈kifejezés〉|
GOTO 〈címke〉{IF 〈kifejezés〉| }

〈azonosító〉 ::= 〈betű〉{〈számjegy〉}1
0

〈betű〉 ::=X |Y |Z
〈számjegy〉 ::=0|1| . . . |9

Az EP nyelv leírása EBNF-fel/2

〈kifejezés〉 ::={+| − | }〈előjeltelen kifejezés〉
〈előjeltelen kifejezés〉 ::= 〈tag〉@{{+|−}〈tag〉}
〈tag〉 ::= 〈tényező〉@{{∗|/}〈tényező〉}
〈tényező〉 ::= 〈azonosító〉|〈szám〉|(〈kifejezés〉)
〈címke〉 ::= 〈szám〉
〈szám〉 ::= 〈számjegy〉{〈számjegy〉}11

0

Fehér szóközök, kocsi vissza/soremelés karakterek.
IF: ha a kifejezés ≥ 0, ugrik.

Példa EP-vel megoldható feladatra
Legnagyobb közös osztó

Írjunk olyan EP programot, mely beolvas két pozitív egész számot,
majd kiírja legnagyobb közös osztóját!

Q: x=x ′>0 y=y ′>0 egészek
R: x=lnko(x ′, y ′)

Ezt megvalósító algoritmus:

x 6= y

\ x > y \ x < y

x := x−y y := y−x

Példa EP-vel megoldható feladatra
EP programlnko-ra

READ X;
READ Y;

100: GOTO 101 IF X-Y;
LET Y=Y-X;
GOTO 100;

101: GOTO 102 IF Y-X;
LET X=X-Y;
GOTO 100;

102: WRITE X;

EBNF használata adatszerkezetek leírására
Bináris fák, listák

Bináris fák: NIL üres objektum

Bináris fa

{
NIL

[gyökér, baloldal, jobboldal]

〈egész bin. fa〉 ::=NIL | [〈egész〉,〈egész bin. fa〉,〈egész bin. fa〉]
〈valós bin. fa〉 ::=NIL | [〈valós〉,〈valós bin. fa〉,〈valós bin. fa〉]

〈egész lista〉 ::=NIL | [〈egész〉,〈egész lista〉]
〈valós lista〉 ::=NIL | [〈valós〉,〈valós lista〉]

EBNF használata adatszerkezetek leírására
W nyelvtan

Módosítás:
〈X̂ lista〉 ::=NIL | [〈X̂ 〉,〈X̂ lista〉]
X̂ := egész|valós|Boole|komplex Hiperszabály

Kétszintű EBNF
van Wijngaarden nyelvtan (W nyelvtan)

〈X̂ bin. fa〉 ::=NIL | [〈X̂ 〉,〈X̂ bin. fa〉, 〈X̂ bin. fa〉]

Szintaxis gráfok
Definicíó

Az EBNF-fel történő leírás szemléletesebb, gráfok segítségével
történő megadása.

• A fogalmak a � jelű csomópontok, bennük a fogalom neve,

• a terminálisok a © jelű csomópontok, bennük egy terminális
sorozat,

• a gráf neve a definiálandó fogalom.

A gráf irányított, összefüggő, egy bemeneti és egy kimeneti éllel.

Formális nyelvek (1. gyakorlat) Formális nyelvek a gyakorlatban 2005/06 II. félév 15 / 18

Szintaxis gráfok
Példa: Azonosító

Szintaxis gráfok
Példa: EP nyelv

Házi feladat

1. Módosított azonosító: belsejében lehet _ jel is. Kezdődhet, de
nem végződhet vele, két aláhúzás nem lehet egymás mellett. Írjuk
fel (E)BNF formulákkal!

2. EP teljes átírása szintaxis gráfokkal.

3. A kiadott PASCAL szintaxis tanulmányozása. A kifejezések
átírása (E)BNF-re.

4. Kifejezések leírása W nyelvtannal.

